

U.S. DEPARTMENT OF THE INTERIOR

U.S. GEOLOGICAL SURVEY

Using the MC-II® Handheld Computer While Running

First or Second Order Leveling Surveys

BY

Kenneth M. Yamashita<sup>1</sup> and Jon M. Yamashita<sup>2</sup>

Open-File Report 92-184

This report is preliminary and has not been reviewed for conformity with U.S. Geological Survey editorial standards (or with the North American Stratigraphic code). Any use of trade, product or firm names is for descriptive purposes only and does not imply endorsement by the U.S. Government.

<sup>1</sup>U.S. Geological Survey  
David A. Johnston Cascades Volcano Observatory  
5400 MacArthur Blvd.  
Vancouver WA. 98661

<sup>2</sup>Mountain Lace Productions  
3905 E. Evergreen Blvd.  
Vancouver, WA. 98661

## TABLE OF CONTENTS

INTRODUCTION .....	3
PROGRAM HISTORY .....	3
MC-II® CONFIGURATION .....	3
GENERAL MC-II® OPERATIONS .....	3
RESETTING THE MC-II® .....	4
PRINTING DATA .....	4
TRANSFERRING FILES OR PROGRAMS WITH KERMIT .....	4
LEVEL PROGRAM OPERATION .....	6
LISTING FOR PROGRAM LEVEL .....	11
EXAMPLE OF A DATA FILE LISTING .....	16
REFERENCES CITED .....	17

## INTRODUCTION

Members of the David A. Johnston Cascades Volcano Observatory typically run First-Order, Class II leveling surveys (Federal Geodetic Control Committee (FGCC), 1984) to monitor volcanoes of the western United States.

The following is a discussion and instruction in the use of the MC-II® handheld computer and the Level program as well as a listing of the Level program codes.

## PROGRAM HISTORY

The original code for program Level was written on a IBM® compatible Personal Computer (PC) using GWBASIC® and downloaded into the MC-II® using the Kermit® software (Kermit is explained later) which came with the MC-II® computer. Although the MC-II® uses MC-BASIC®, and not GWBASIC®, the codes are similar enough that in this program no changes in the code were required.

Because no software is available for this type of leveling survey, we developed our own software. We chose the MC-II® computer for our field data collector.

The Level program requires about 5 kilobytes (KB) of Random Access Memory (RAM), plus an additional 10 KB is required for storage of a day's worth of data.

## MC-II® CONFIGURATION

The MC-II® came with 128 KB of RAM. We added an additional 128 KB RAM module for a total of 256 KB. However only 128 KB is available for the program and data storage; the rest is used for the operating system. If peripherals will be used with the MC-II® computer the communication package should also be purchased.

One 128 KB RAM module was installed in the "A" drive and the other was installed in the "B" drive (figure 2). The "A" drive was configured to "Memory", for the operating system, the "B" drive was configured to "Files" for programs and data storage, and the MC-BASIC® module was installed in the "E" drive.

The two, 128 KB RAM modules contain their own lithium batteries, so the modules can be removed without losing files or programs contained in them. The MC-BASIC module is an Erasable, Programmable, Read Only Memory (EPROM) module, so its memory is always protected.

## GENERAL MC-II® OPERATIONS

**Note:** The [↵] symbol means "Enter". This key is also marked "TAB".

To clear the screen type "CLS" then the [↵] key.

To access the clock and calendar functions type "TIME" then [↵] or "DATE" then [↵].

To exit press the [↵] key twice.

When "Filename" and "Programname" are referred to in the text, this means that the actual filename, or program name is to be entered.

These commands will not work while in the "BASIC" mode.

The MC-II® can be turned off in two ways. The easiest is to type "OFF", or "EXIT". From the main menu or while in BASIC use the [F5], (OFF) key.

When the computer is first turned on it will "boot up" in the "F" drive and a menu will be displayed. Press the [↵] key. Another menu will appear. The first line displayed will

be "Current dev = F:". Use the "up" [▲] (PG UP) key to highlight "B:", then press the [F1] (CD) key. The computer will now be in drive B. Type **BASIC**. This will activate MC-BASIC. Activate the "CAPS" mode by pressing the [white] (SH1) key located on the right middle of the keyboard then the [O] (CAPS) key. Load the program by pressing the [F3] (LOAD) key. The message {LOAD} will appear. Type the **PROGRAM NAME** then [↵]. The message {OK} will appear. Press the [F2], (RUN) key. The program is now activated.

These commands can also be initialized by typing **Load**, followed by the program name, and **RUN**, then the program name, and ending the input with the [↵] key.

To access the numeric keys press the [white] (SH1) key then the [N], (NUM) key located at the middle left of the keyboard. The MC-II® will remain in the numeric mode until the [SH1] and [N], (NUM) keys are pressed again, or by pressing the [ON] key, which resets the keyboard. While in the numeric mode, alpha characters shared with the numeric keypad can be accessed by pressing the [SH1] key then the alpha character. Alpha characters not contained within the numeric keypad need not be preceded by the [SH1] key, however, those characters such as \*, /, @ etc. will require the [SH1] key first. Once the alpha character is entered the MC-II® automatically goes back into the numeric mode. The [SH1] key needs to be pressed before entering each alpha character shared with the numeric keypad unless the numeric key pad has been disengaged as explained earlier.

Unwanted characters can be erased with the [←] key located in the middle of the keyboard. The [◀] or [▶] cursor keys will move the cursor left or right without overwriting or deleting characters. The cursor keys should not be used when a program is running as this can cause run-time errors. To exit basic, type **SYSTEM**, then the [↵] key.

### RESETTING THE MC-II®

In the event that the MC-II® keyboard or screen "lock up", a "soft" reset can be performed by depressing the [F1] and the [ON] keys simultaneously. The [F2] and (ON) and the [F3] and (ON) keys are higher levels of "reset". If the soft reset fails, open the battery compartment of the MC-II®; at the upper left hand corner of the battery compartment is a 2 mm silver button, use a small object such as a ball point pen and depress the button, this will cause a "hard" reset. No stored programs or files will be lost by this procedure.

### PRINTING DATA

To get a paper copy or a "hard copy" of the program file or of the data set, connect the printer to Com port 2 (communication port) located at the top left of the MC-II® as it faces upright (figure 1). Type "**COPY FILENAME COM2:**", followed by the [F1], (GO) key. Another way is to highlight the file name using the ▲, or ▼ cursor key, then press the [F2], (copy) key. The MC-II® will prompt {Destination?}. Type "**COM2:**", followed by the [↵] key.

### TRANSFERING FILES OR PROGRAMS WITH "KERMIT"®

*Kermit is copyrighted by Columbia University, NY., but permission is granted for free distribution by the trustees of the University.*

Kermit will need to be installed on both the MC-II®, and the PC.

### Transfer from the MC-II® to a PC

If your PC has a hard drive, go to step "e".

To transfer files or programs from the MC-II® to a PC without a "hard" drive:

- a) Insert the Disk Operating System (DOS) Boot disk into the "A" drive of the PC.
- b) Insert the data diskette into the "B" drive.
- c) Turn the computer on.

After the computer has "Booted" up, remove the Boot disk from the "A" drive.

- d) Insert the diskette containing the Kermit® program into the "A" drive.
- e) Using the serial adapter provided with the communication package, connect the serial connector to the serial port of the PC. The other end of the "Pigtail" cable goes to Com port 1 located on the right-hand side of the MC-II® as it faces upright.
- f) On the PC type "Kermit r b:" then the [↵] key. DO NOT ENTER THE QUOTE (") MARKS!!
- g) On the MC-II® type "Kermit s filename" then the [↵] key. The "r" and "s" after Kermit stands for "receive" and "send".
- e) While the data are being transferred, the PC will indicate the number of "packets" received and the MC-II® will display the number of "bytes" sent. You have ten (10) tries before either computer aborts the transmit or receive sequence.

### Transfer from PC to MC-II®

If sending files or programs from the PC to the MC-II®, the only difference would be switching the "r" and the "s" after Kermit on the PC and the MC-II®.

## LEVEL PROGRAM OPERATION

NOTE: All alpha characters MUST BE ENTERED IN CAPS.

Characters italicized and enclosed in { } are program prompts or messages. Characters enclosed in " " are user input characters. ##.# or XXX will be replaced by "real" numbers or characters when the program is running. Terminate all input with the [↵] key.

To run the level program first load the program by pressing the [F3] (LOAD) key. The message {LOAD} will appear. Type "LEVEL" then [↵]. The message {OK} will appear. Press the [F2], (RUN) key. The program Level is now activated.

1) {*Date is XX-XX-XX*} The year, month and day will be displayed. If this is correct enter "Y", if not, enter "N". If "N" is entered the next prompt will be:  
{*is this OK?*}

1a) {*Enter the new date mo-da-yr*} Enter the new date.

If "Y" was entered for step number 1 the program would prompt:

2) {*Time is XX:XX:XX*} The time will be displayed. If the time is OK enter a "Y", if not enter "N". If "N" is entered the next prompt will be:  
{*is this OK Y, N?*}

2a) {*Enter the new time hr:min:sec*} Enter the new time.

3) {*Enter File Name?*} Enter a unique name. If a file with a similar name exists, the next prompt will be:

3a) {*File XXXX*} This allows you to create a new file or to overwrite or  
{*already exists*} append to an existing file.

{*do you want to*}

{*(O) overwrite*}

{*(A) append or*}

{*(C) create a new*} If "A" is entered, the program appends the file  
If "C" is entered, the program cycles back to step # 3

{*file? O, A, C*}

If "O" is entered, the next prompt will be:

3b) {*Do you really*} If "O" was entered, this prompt will be generated  
{*mean overwrite? Y,N*} If "N" is entered, the program will cycle back to step 3.

4) {*1st or 2nd Order?*} 1st order surveys (FGCC,1984) allow 0.3 mm error between the low and high scale elevations. 2nd order allows 1.0 mm. Enter

"1" or "2".

- 5) {*Do you want stadia Y, N*} If "Y" is entered, all readings will require stadia wire readings, a "N" will not ask for stadia wire reading.
- 6) {*Enter Gun ID*} Enter the serial number of the level instrument.
- 7) {*Level Type*} Enter the type and manufacturer of the level, ie; "WILD NA2".
- 8) {*Rod Type (W) WILD*} Enter the manufacturer of the level rods.  
{*(K) KERN*}
- 9) {*Enter Network*} Enter the location of the survey. Not more than 10 characters.
- 10) {*Gun HI in FT*} Enter the height of the gun person.
- 11) {*Enter both rod ID*} Enter both rod ID numbers separated by a space.
- 12) {*BS Bench mark*} Enter the ID of the backsight bench mark.
- 13) {*BS rod ID*} Enter the serial number of the rod on the backsight bench mark.

If "Y" was given to prompt number (5) the next prompt will be:

- 14) {*STADIA IMBALANCE IS ##.# M.*} The first time the program is run it will  
{*IS THIS CORRECT Y,N*} always be 0. If the number displayed is correct, enter a "Y". If not, enter a "N".

If an "N" was entered the prompt will be:

- 15) {*Enter the correct imbalance*} How do you know what the correct imbalance is?  
Step (30) will display the total imbalance.

If "Y" was given for step number (14), the prompt would have been:

- 16) {*enter 1 to change*} Entering a 1 at any of the middle wire inputs will produce  
{*or correct input*} a menu to change program input sequence.
- 17) {*BS mw low scale*} Normally at this point the backsight (BS) middle wire (mw) low scale will be read, but if the previous foresight was a bench mark, entering "1" allows you to input a bench mark ID.

If "N" was given for step number (5), the next two prompts will not appear but the program

will cycle to step number (20) *{FS mw low scale}*.

18) *{BS top wire}* Enter backsight top stadia wire.

19) *{BS Bott wire}* Enter backsight bottom stadia wire .

A "BEEP" at this point means that there was an error in the input or in the numbers read. The program will cycle back to number (18).

If there is an error again, a double "BEEP" will sound and the program will cycle to step number (17).

If there are no errors, the message *{BS Stadia is ###.# m}* will be displayed, then the prompt:

20) *{FS mw low scale}* Normally the foresight (FS) middle wire low scale reading will be entered but, if "1" is entered, a menu will be generated.

If "N" was given at step number (5), the program would cycle to step number (25), bypassing steps 21-24. If "Y" was entered, the following prompt will be displayed:

21) *{FS Top Wire}* Enter foresight top stadia wire.

22) *{FS Bott Wire}* Enter foresight bottom stadia wire.

Again "Beeping" sounds require additional input. Read the prompts carefully!!.

If the stadia imbalance between the backsight and the foresight is greater than  $\pm 5$  m (FGCC, 1984), one of the following two messages will be displayed:

23) *{Foresight is long by ###.# m}* **OR** *{Foresight is short by ###.# m}*

*{Move Someone!!}* Inform the gun person of this amount so the proper adjustments can be made. The program will then cycle to step number (17).

17) *{BS mw low scale}* Read the setup again. DO NOT ENTER a "1"!!.

If everything had checked at step number (22), the following message will be displayed:

24) *{FS Stadia is ###.# m}*

25) *{Foresight mw high}* Enter the foresight high, middle wire.

26) *{Backsight mw high}* Entering a "1" will generate a menu



A "BEEP" at this point requires immediate attention by the recorder. One of two messages will be displayed. If the backsight high reading is less than 302, the computer will prompt *{Read the High Scale!!}*. This means that the operator read the low scale instead of the high scale. This allows the operator a second chance to correct the mistake. If the message is *{Warning Bad Reading}* the program will display *{Check is off by ###.##}* and cycle back to step number (17) *{BS mw low scale}*. Since the stadia readings were good the first time they were read, the program will not ask for the stadia readings again.

If there are a number of "Bad Readings" and the gun person wishes to move the instrument, and or the rod people, you must enter "1" for step number (17). This will generate a menu. Enter 5 "Reject stadia" reading for the set-up. This should be the only time that a "5" from the menu is used!!!

If the elevation difference between the low and high scale is within the tolerance set by step number (4), the prompt will be:

27) *{Enter temp top then bott}* Enter both temperatures as one number. A temperature reading of 23.5 on the top and 24.7 on the bottom will be entered as 235.247. Temperatures of 10.3 and 9.7 are entered as 103.097

At this point a number of messages will be displayed on the screen.

28) *{Stadia Distance is ###.# m}* This is informational only and is the cumulative distance since the last bench mark. This is helpful when approaching the required distance to the next bench mark.

29) *{Elev is ###.## m}* Again informative information but useful when approaching a bench mark.

The next message displayed will be either:

30) *{Lengthen} OR* This number is the distance required to keep the length of the backsight stadia and the foresight equal. Inform the rod people so they can correct accordingly. The program will then cycle back to step (17).  
*{Shorten}*  
*{the foresight ###.##}*

17) *{BS mw low scale}* Continue as before. If the foresight rod was on a bench mark enter "1" now.

If "1" is entered at step number (17) a menu will be generated. Enter "1" for "Input a BM"

31) *{FS BM ID}* Enter the foresight bench mark ID number. If you made a mistake and

did not want to enter a bench mark, enter a "1" again. Program will return to step 17.

- 32) {*% of cloud cover*} Enter the percentage of cloud cover that the survey was conducted under or enter the NGS code. 0 if less than 25%, 1 for 25-75% and 2 if more than 75% cloud cover.
- 33) {*Wind direction and speed*} Enter this as "NE 10" if from the northeast at 10 mph or enter the NGS code 1 for wind speed of less than 5 MPH, 2 for wind speeds 5-15 MPH or 3 if greater than 15 MPH.
- 34) {*Elev is ###.## m*} This is the total elevation difference from the previous bench mark.
- 35) {*Total Dist is ##.## m*} This is the total distance from the last bench mark.
- 36) {*Cont or E?, C, E*} Be careful!!. Entering a "E" will end the program. Entering a "C" to continue, will display:
- 37) {*Was FS a BM or TBM?*} Ask your gun person!!. Entering a BM (bench mark) will zero everything, elevations, stadia, all. Entering a TBM (temporary bench mark) keeps everything and subsequent values are accumulated.

If a "BM" was entered, the program will cycle back to step number (6).

6) {*Enter Gun ID*}.

If a "TBM" was entered, the program will cycle to step number:

12) {*BS bench mark*} Enter the backsight bench mark and continue as before.

## LISTING FOR PROGRAM LEVEL

```
10 REM PROGRAM "LEVEL"
20 REM VERSION 3.5
30 REM WRITTEN BY KENNETH M. AND JON M. YAMASHITA, MAY 1991
40 KEY OFF
50 PRINT "DATE IS "
60 PRINT DATE$
70 INPUT "IS THIS OK Y, N ";ANS$
80 IF ASC(ANS$)>90 THEN BEEP:PRINT "ENTER INPUT IN CAPS":GOTO 70
90 IF ANS$ ="Y" THEN 150
100 PRINT "ENTER NEW DATE"
110 INPUT "MO-DA-YR",DAY$
120 DATE$=DAY$
130 PRINT DATE$
140 GOTO 50
150 PRINT "TIME IS "
160 PRINT TIME$
170 INPUT"IS THIS OK, Y, N ";ANS$
180 IF ASC(ANS$)>90 THEN BEEP:PRINT "ENTER INPUT IN CAPS":GOTO 150
190 IF ANS$ = "Y" THEN GOTO 240 ELSE
200 PRINT "Enter new time"
210 INPUT"HR:MIN:SEC",HR$
220 TIME$=HR$
230 GOTO 150
240 INPUT"Enter filename ";FILENAME$
250 ON ERROR GOTO 420
260 OPEN "I", #3, "b:" + FILENAME$
270 BEEP:PRINT FILENAME$
280 PRINT "Already exists"
290 PRINT "do you want to"
300 PRINT "(O) Overwrite"
310 PRINT "(A) Append or"
320 PRINT "(C) Create a new"
330 INPUT "file? O, A, OR C ";FILE$
340 IF FILE$ = "A" THEN GOTO 380
350 IF FILE$ = "O" THEN GOTO 390
360 IF FILE$ = "C" THEN CLOSE #3 :GOTO 240
370 PRINT "No such choice" :BEEP :GOTO 290
380 CLOSE #3 :OPEN "B:" + FILENAME$ FOR APPEND AS #3 :GOTO 440
390 BEEP :PRINT "Do you really mean"
400 INPUT "overwrite Y, N ";OVERWRITE$
410 IF OVERWRITE$<>"Y" THEN CLOSE #3: GOTO 240
420 CLOSE #3
430 OPEN "O", #3,"B:" + FILENAME$
440 DIM RUNNINGELEV(300)
450 RUNNINGELEV(1) = 0 :X = 0: CORRIMBALANCE = 0
460 INPUT"1 or 2 order 1, 2 ";ORDER
470 IF ORDER <> 1 AND ORDER <> 2 THEN :PRINT "NO SUCH CHOICE" :BEEP:GOTO 460
480 IF ORDER = 1 THEN G = .03 ELSE G = .1
490 INPUT "want stadis Y, N ";STADIA$
```

```

500 IF STADIA$ <> "Y" AND STADIA$ <> "N" THEN BEEP :PRINT "NO SUCH
CHOICE";STADIA$ :GOTO 490
510 IF STADIA$ = "N" THEN STADIA$ = "NO" ELSE STADIA$ = "YES"
520 INPUT "Enter Gun ID ";GUNID$
530 INPUT "Level Guntype ";LEVELTYPE$
540 INPUT"Temp units C, F ";UNIT$
550 PRINT "Rods Type (W) Wild
560 INPUT "(K) Kern ";RODTYPE$
570 IF RODTYPE$="W" THEN RODTYPE$="WILD" ELSE RODTYPE$="KERN"
580 INPUT "Enter Network ";NETWORK$
590 INPUT "Gun HI in ft ";GUNHI$
600 PRINT "Enter Both Rod ID"
610 INPUT "Separate by Space ";RODID$
620 INPUT "BS Bench Mark ";BSBM$
630 INPUT "BS Rod ID ";BSRODID$
640 IF STADIA$="NO" THEN GOTO 700
650 PRINT USING "I_m_b_a_l_a_n_c_e_i_s_###.#_m";STADIAIMBALANCE
660 INPUT"is this OK Y,N ";CORRSTAD$
670 IF CORRSTAD$="Y" THEN GOTO 700
680 INPUT"Correct imbalance ";CORRIMBALANCE
690 STADIAIMBALANCE = 0
700 PRINT #3,DATE$
710 PRINT #3,TIME$
720 PRINT #3,GUNID$
730 PRINT #3,LEVELTYPE$
740 PRINT #3,UNIT$
750 PRINT #3,RODTYPE$
760 PRINT #3,NETWORK$
770 PRINT #3,GUNHI$
780 PRINT #3,RODID$
790 PRINT #3,BSBM$
800 PRINT #3,BSRODID$
810 BSTADIA = 0
820 FSTADIA = 0
830 READSTADIA$ = STADIA$
840 X=X+1
850 BEEP
860 PRINT "Enter 1 to change"
870 PRINT "or correct input"
880 INPUT"BS MW Low";BSLOW
890 IF BSLOW = 1 THEN GOSUB 2190:GOTO 880
900 I=0
910 I=I+1
920 IF READSTADIA$="NO" THEN 1020
930 IF I>2 THEN BEEP : GOTO 880
940 INPUT "BS Top Wire";BSTOPWIRE
950 INPUT "BS Bottom Wire";BSBOTWIRE
960 BSTADIAINTERVAL1 = BSTOPWIRE-BSLOW
970 BSTADIAINTERVAL2 = BSLOW-BSBOTWIRE
980 BSTADIACHECK = BSTADIAINTERVAL1 - BSTADIAINTERVAL2
990 IF ABS(BSTADIACHECK) >.5 THEN BEEP : GOTO 910
1000 BSTADIA = BSTADIAINTERVAL1 + BSTADIAINTERVAL2

```

```

1010 PRINT USING " B S s t a d i a i s _ ##.# _ m";BSTADIA
1020 INPUT "FS MW LOW";FSLOW
1030 IF FSLOW = 1 THEN BSTADIA=0: GOSUB 2190:GOTO 1020
1040 I=0
1050 I=I+1
1060 IF READSTADIA$ = "NO" THEN 1410
1070 IF I>2 THEN BEEP : GOTO 1020
1080 INPUT "Fs Top Wire";FSTOPWIRE
1090 INPUT "Fs Bottom Wire";FSBOTWIRE
1100 FSTADIAINTERVAL1 = FSTOPWIRE - FSLOW
1110 FSTADIAINTERVAL2 = FSLOW - FSBOTWIRE
1120 FSTADIACHECK = FSTADIAINTERVAL1 - FSTADIAINTERVAL2
1130 IF ABS(FSTADIACHECK) > .5 THEN BEEP : GOTO 1050
1140 FSTADIA = FSTADIAINTERVAL1 + FSTADIAINTERVAL2
1150 STADIACHECK=BSTADIA - FSTADIA
1160 IF ABS(STADIACHECK) < 5 THEN GOTO 1240 ELSE BEEP
1170 PRINT " MOVE SOMEONE !!" :READSTADIA$ = "YES"
1180 PRINT "Foresight is"
1190 IF STADIACHECK >0 THEN GOTO 1220
1200 PRINT USING "_ l o n g _ b y _ ##.# _ m e t e r _ s";ABS(STADIACHECK)
1210 GOTO 860
1220 PRINT USING "_ s h o r t _ b y _ ##.# _ m e t e r _ s";STADIACHECK
1230 GOTO 860
1240 TOTALBSTADIA = TOTALBSTADIA + BSTADIA
1250 TOTALFSTADIA = TOTALFSTADIA + FSTADIA
1260 STADIAIMBALANCE = TOTALFSTADIA -TOTALBSTADIA + CORRIMBALANCE
1270 IF ABS(STADIAIMBALANCE) < 5 THEN GOTO 1400
1280 BEEP : PRINT "Stadia imbalance is"
1290 PRINT"off by ";STADIAIMBALANCE
1300 PRINT"Foresight is"
1310 IF STADIAIMBALANCE >0 THEN GOTO 1340
1320 PRINT USING "_ l o n g _ b y _ ##.# _ m e t e r _ s";ABS(STADIAIMBALANCE)
1330 GOTO 1350
1340 PRINT USING "_ s h o r t _ b y _ ##.# _ m e t e r _ s";STADIAIMBALANCE
1350 TOTALBSTADIA = TOTALBSTADIA - BSTADIA
1360 TOTALFSTADIA =TOTALFSTADIA-FSTADIA
1370 STADIAIMBALANCE = STADIAIMBALANCE - (TOTALFSTADIA - TOTALBSTADIA)
1380 PRINT " MOVE SOMEONE !!" :READSTADIA$ = "YES"
1390 GOTO 850
1400 PRINT USING " F S s t a d i a i s _ ##.# _ m";FSTADIA
1410 INPUT "FS MW High";FSHIGH
1420 IF FSHIGH = 1 THEN GOSUB 2190:GOTO 1410
1430 INPUT "BS MW High";BSHIGH
1440 IF BSHIGH = 1 THEN GOSUB 2190:GOTO 1430
1450 IF BSHIGH <302 THEN BEEP :PRINT "Read the high scale!!" :GOTO 1430
1460 LOWSCALELEV=BSLOW-FSLOW : HIGHSCALELEV = BSHIGH - FSHIGH
1470 AVELEVDIFF =(LOWSCALELEV + HIGHSCALELEV)/2
1480 ELEVCHECK = LOWSCALELEV - HIGHSCALELEV
1490 IF ABS(ELEVCHECK) < G THEN GOTO 1540
1500 BEEP : BEEP : PRINT "WARNING BAD READING "
1510 PRINT "CHECK IS OFF BY ";
1520 PRINT USING"### _ c _ m";ELEVCHECK

```

```

1530 READSTADIA$ = "NO" : GOTO 850
1540 AVELEVDIFF =(LOWSCALELEV + HIGHSCALELEV)/2
1550 INPUT"Temp top.bott";TEMP$
1560 IF STADIA$="NO" THEN GOTO 1600
1570 STADIADISTANCE = STADIADISTANCE +BSTADIA + FSTADIA
1580 PRINT "Stadia Distance is "
1590 PRINT USING "####.#_ _m_e_t_e_r_s";STADIADISTANCE
1600 PRINT #3,BSLOW
1610 PRINT #3,BSHIGH
1620 PRINT #3,USING "##.#";BSTADIA
1630 PRINT #3,FSLOW
1640 PRINT #3,FSHIGH
1650 PRINT #3,USING "##.#";FSTADIA
1660 PRINT #3,TEMP$
1670 BSTADIA = 0: FSTADIA = 0
1680 RUNNINGELEV(X) = AVELEVDIFF + RUNNINGELEV(X-1)
1690 PRINT USING"E l e v _ i s _###.###_ m";RUNNINGELEV(X)/100
1700 PRINT #3,USING "#####.#####";RUNNINGELEV(X)
1710 IF STADIA$="NO" THEN GOTO 830
1720 IF STADIAIMBALANCE>0 THEN GOTO 1740
1730 PRINT USING"L_e_n_g_t_h_e_n_ F S _#.#_ m";ABS(STADIAIMBALANCE) : GOTO 08
1740 PRINT USING"S_h_o_r_t_e_n_ F S _#.#_ m";ABS(STADIAIMBALANCE):GOTO 830
1750 INPUT "WANT TO END!Y,N ";ENDI$
1760 IF ENDI$ <> "Y" THEN E5$="C": GOTO 1980
1770 PRINT "PROGRAM ENDED"
1780 CLOSE #3
1790 KEY ON
1800 END
1810 BEEP
1820 INPUT "FS BM ID ";FSBM$
1830 IF FSBM$ = "1" THEN BEEP:GOTO 850
1840 PRINT #3,FSBM$
1850 IF STADIA$="NO" THEN GOTO 1880
1860 PRINT #3,USING"####.#";STADIADISTANCE
1870 PRINT #3,USING"##.#";STADIAIMBALANCE
1880 INPUT"% Cloud cover ";CLOUD$
1890 PRINT "Wind speed "
1900 INPUT"and direction ";WIND$
1910 PRINT #3,CLOUD$
1920 PRINT #3,WIND$
1930 PRINT USING"E l e v _ i s _###.###_ m ";RUNNINGELEV(X-1)/100
1940 IF STADIA$="NO" THEN GOTO 1970
1950 PRINT USING"T o t a l _ D i s t _#####.#_ m";STADIADISTANCE
1960 IF STADIAIMBALANCE>0 THEN GOSUB 2010 ELSE GOSUB 2040
1970 INPUT "Cont Or End? C, E ";E5$
1980 IF E5$="E" THEN 1750 ELSE INPUT "Was FS a BM or TBM";S5$
1990 IF S5$="TBM" THEN X=X-1: GOTO 620 ELSE RUNNINGELEV(X-1)=0
2000 TOTALBSTADIA=0 : STADIAIMBALANCE = 0 : TOTALFSTADIA = 0
:STADIADISTANCE=0:GOTO 520
2010 PRINT "Foresight is"
2020 PRINT USING"##.#_ _m_e_t_e_r_s_ _l_o_n_g";STADIAIMBALANCE
2030 RETURN

```

```

2040 PRINT "Foresight is"
2050 PRINT USING"#.#_ _m_e_t_e_r_s_l_o_n_g";ABS(STADIAIMBALANCE)
2060 RETURN
2070 BEEP
2080 PRINT "ARE YOU SURE YOU"
2090 PRINT "WANT TO REJECT"
2100 INPUT "THESE STADIA?Y,N",REJECT$
2110 IF REJECT$<> "Y" THEN GOTO 850
2120 TOTALBSTADIA = TOTALBSTADIA - BSTADIA
2130 TOTALFSTADIA = TOTALFSTADIA - FSTADIA
2140 STADIAIMBALANCE = STADIAIMBALANCE - (TOTALFSTADIA - TOTALBSTADIA )
2150 BEEP: PRINT "Stadias have been"
2160 PRINT "Rejected, re-read all"
2170 PRINT "stadias and rod readings"
2180 X=X-1 : GOTO 830
2190 BEEP
2200 PRINT "1) INPUT A B M ?"
2210 PRINT "2) INPUT BS LOW ?"
2220 PRINT "3) INPUT FS LOW ?"
2230 PRINT "4) INPUT FS HIGH ?"
2240 PRINT "5) REJECT STADIAS ?"
2250 PRINT "6) EXIT MENU ?"
2260 INPUT "ENTER NUMBER"; CHOICE
2270 IF CHOICE = 1 THEN GOTO 1810
2280 IF CHOICE = 2 THEN GOTO 880
2290 IF CHOICE = 3 THEN GOTO 1020
2300 IF CHOICE = 4 THEN GOTO 1410
2310 IF CHOICE = 5 THEN GOTO 2070
2320 IF CHOICE = 6 THEN RETURN
2330 PRINT "NO SUCH CHOICE" :BEEP:GOTO 2200

```

## EXAMPLE OF A DATA FILE LISTING

<u>Stored data</u>	<u>Explanation</u>
04-02-1991	Month, Day and the Year of the survey
15:26:15	Time in military format
123456	Serial number of the level instrument
WILD NA2	Manufacturer and type of level instrument
C	Temperatures units
WILD	Manufacturer of the level rods
YELLOWSTONE	Network
5.8	Height of gun person in feet
5140A 5140B	Rod ID of both rods
CV089-450	Backsight bench mark ID
5140B	Rod ID on the backsight bench mark
259.3	Backsight rod low scale reading
560.95	Foresight rod low scale reading
15.2	Stadia distance from the instrument to the backsight rod
49.355	Foresight rod low scale reading
351.000	Foresight rod high scale reading
14.6	Stadia distance from the instrument to the foresight rod
698.673	Temperature, if taken. This example shows 69.8 and 67.3
209.9475	Elevation difference for set up in centimeters
CV089-455	Foresight bench mark
29.8	Cumulative stadia distance from the last bench mark
-.6	Stadia imbalance at the bench mark
10%	Percentage of cloud cover. Use when code is not known
NE12	Direction and wind speed. Again when code is not known

\*\*\* In most cases the data would be printed at a bench mark but if the decision is to continue the survey, the following output will be generated. If a bench mark had not been encountered, the next data point in the file would have been the backsight low scale reading.

```
04-02-1991
15:28:04
123456
WILD NA2
WILD
YELLOWSTONE
5.8
5140A 5140B
CV089-455
5140A
49.355
351
14.6
259.3
560.95
15.2
123.123
0
```



## REFERENCES CITED

Federal Geodetic Control Committee, 1984, Standards and specifications for geodetic control networks: Rockville, Maryland, National Oceanic and Atmospheric Administration.

Figure 1




Figure 2

